

Paper: 056-2010

## Two Practical Approaches to Add Special Characters to Axis Labels in SAS/Graph Output in RTF

Amos Shu, INC Research, Inc., King of Prussia, PA

### ABSTRACT

There are several methods (such as the BYTE function, Hexadecimal code, and Greek character symbols with/without the ODS ESCAPECHAR= statement) available to produce special characters (such as superscripts three, Greek letters, and symbols) in SAS. However, none of them can be used directly in the AXIS statement to add special characters in axis labels in SAS 8.2. This paper discusses how to increase the flexibility of your program by using 1) TITLE/FOOTNOTE statements to add special characters in the vertical axis label and 2) the ANNOTATE FACILITY to add special characters in the horizontal axis label.

### INTRODUCTION

The easiest way to get special characters is via the Character Map system tool in Windows or using specific keystroke sequences such as Alt+0174. The BYTE function<sup>[1]</sup> or Hexadecimal code with/without the ODS ESCAPECHAR= statement can also be used to insert special characters into output reports. Using the DDE technology and MS WORD formatting is another way to get special characters<sup>[2]</sup>. A fourth approach is changing font type to the Greek<sup>[3]</sup>. Unfortunately, none of those methods can be used in the AXIS statement to directly insert special characters into the axis labels of SAS graphics in RTF.

This paper first discusses the use of the TITLE statement in the vertical axis labels (including right vertical axis) with special characters in SAS/Graph output RTF destination, then the use of the FOOTNOTE statement in labeling of a horizontal axis with special characters, and finally the use of the ANNOTATE DATA SET to add special characters on the horizontal axis. All programs presented in this paper were developed in PC SAS 8.2 in the Windows environment.

### USING THE TITLE STATEMENT TO ADD SPECIAL CHARACTERS TO VERTICAL AXIS LABELS IN SAS/GRAPH

The following program illustrates how the TITLE statement is used to display the label with special characters in the vertical axis. OPTION a=90 in TITLE2 positions TITLE2 at the left edge of the graphics output area. OPTION move = (-0.5,+0) positions the superscript three. Appendix 1 shows the output.

```
data xyz;
  input treatment visit $ mean ;
  cards;
1 visit1 147
2 visit1 250
1 visit2 165
2 visit2 515
1 visit3 166
2 visit3 595
1 visit4 195
2 visit4 1045
1 visit5 426
2 visit5 267
1 visit6 212
2 visit6 780
;

/* generate graph */
options nodate orientation=landscape;
goptions reset=all device=png target = png gunit=pct
         cback=white colors=(black) ftext=swiss
         rotate=landscape
         xmax=8in ymax=6in
         xpixels=5400 ypixels=3600;

title1 h=2 " ";
title2 j=c a=90 r=0 h=4 "ABC Count (/mm" move=(-0.75,+0)
      h=3 '3' move=(+0.75,+0) h=4 ')';
```

```

title3 j=c h=5 "Figure 123: ABC Count" ;
title4 h=2 " " ;

footnote h=6 " " ;

/* customize Axis */
axis1 MAJOR=(h=1) MINOR=none OFFSET=(2,2) WIDTH =3
      LABEL=(h=4 j=c "Visit") VALUE = (h=2.5) width=3;

axis2 MAJOR=(h=1) MINOR=(number=5 h=.5) OFFSET=(0,0) WIDTH =3
      ORDER=(0 to 1200 by 100) LABEL=none VALUE=(h=2.5);

symbol1 i=j c=black v=dot line=1 w=6 h=2 ;
symbol2 i=j c=black v=triangle line=2 w=6 h=2 ;

proc gplot data=xyz ;
  plot Mean*visit = treatment / haxis=axis1 vaxis=axis2 nolegend;
run;
quit;

```

Simply adding more TITLE statements with OPTION a =90 and different title line values can add more labels on the same vertical axis if there is a need in real practice. To display a label on the right vertical axis, just change the OPTION to a=-90 with a different title line value. For example, `Title6 j=c a=90 r=0 h=4 "ABC2 Count (/mm" move=(-0.75,+0) h=3 '3' move=(+0.75,+0) h=4 ' ');`

Similar to the TITLE statement, using FOOTNOTE statements can also achieve the same results with vertical axis labels.

As an aside, In-line formatting does not seem to work for SAS graphic output. The following statement did not generate the expected results.

```

ods escapechar = '^';
title3 j=c a=90 r=0 h=4 "ABC Count (/mm" ^{super 3} " )";

```

## USING THE FOOTNOTE STATEMENT TO ADD SPECIAL CHARACTERS TO HORIZONTAL AXIS LABELS IN SAS/GRAPH

Is it possible to use footnote statement to label horizontal axis with special characters? For instance, `footnote1 j=c h=4 "Dose (" "E4"X "g/kg)"` will display "Dose ( $\mu\text{g}/\text{kg}$ )". However, even with FOOTNOTE1, which positions the footnote immediately outside the output area, it still appears slightly too far from the normal labeling area. It is not recommended to use the first FOOTNOTE statement to label horizontal axis if other methods are available.

## USING THE ANNOTATE FACILITY STATEMENT TO ADD SPECIAL CHARACTERS TO HORIZONTAL AXIS LABELS IN SAS/GRAPH

An ANNOTATE DATA SET can be used to label the horizontal axis with special characters. The following program illustrates how the second label function generates superscript three and inserts it into the text `'ABC Count (/mm )'` (there is a blank space here between "mm" and the close parenthesis to leave space for the superscript three). The resulting plot is in Appendix 2.

```

/* Create annotate dataset to label axis */
data anno;
  length function $8 text $50 ;
  xsys='5'; ysys='5'; when='A';
  function='label'; text='ABC Count (/mm )'; size=7.5; position='1'; X=60; Y=-
6;
  output;
  function='label'; text='3'; size=5.5; position='1'; X=59.4; Y=-4;
  output;
run;

title1 h=2 " " ;
title2 j=c h=5 "Figure 123: ABC Count" ;
title3 h=2 " " ;

```

```

footnote      h=6 " " ;

              /* customize Axis */
axis2 MAJOR=(h=1) MINOR=none OFFSET=(2,2) WIDTH =3
      LABEL= (h=3 j=c a=90 "Visit") VALUE = (h=2.5);

axis1 MAJOR=(h=1) MINOR=(number=5 h=.5) OFFSET=(0,0) WIDTH =3
      ORDER=(0 to 1200 by 100) LABEL=none VALUE=(h=2.5);

symbol1 i=j c=black v=dot line=1 w=6 h=2 ;
symbol2 i=j c=black v=triangle line=2 w=6 h=2 ;

proc gplot data=xyz ;
  plot visit*Mean = treatment / haxis=axis1 vaxis=axis2 nolegend annotate =
anno;
run;
quit;

```

## CONCLUSION

Special characters are sometimes necessary in SAS output and SAS/Graph output. While there is no way to produce special characters by directly using AXIS statement, the TITLE/FOOTNOTE statements can be easily used to label vertical axis with special characters in SAS graphics in RTF, and ANNOTATE DATA SETS can do the same for horizontal axis.

## ACKNOWLEDGEMENTS

A Special thanks to Bari Lawhorn of SAS and Thomas Kinghorn for their help.

## REFERENCES

- [1]. Quan Ren, *Using Special Characters and Formats with SAS*, Pharmasug, 2001.
- [2]. Cindy Song, *Producing Special Characters in SAS Output in RTF*, Pharmasug, 2000.
- [3]. Doug Lassman, *Fonts and Special Characters in SAS – It's All Greek to Me*, NESUG, 2003.

## CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Amos Shu  
 INC Research, Inc.  
 2200 Renaissance Blvd.  
 King of Prussia, PA 19406  
 Email: [hshu@incresearch.com](mailto:hshu@incresearch.com)


## TRADEMARK INFORMATION

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. © indicates USA registration.

Other brand and product names are trademarks of their respective companies.

Appendix 1 – Vertical Axis Label with Superscript Three

Figure 123: ABC Count



Appendix 2 – Horizontal Axis Label with Superscript Three

Figure 123: ABC Count

